



## 30 Tabellenblattereignisse (Change, BeforeDoubleClick, ...)

## Ausgangssituation:

- Wir starten mit dem unten dargestellten Tabellenblatt. Dort befinden sich fiktive Werte, anhand derer wir die Beispiele durchgehen.

|   | A | B      | C      | D      | E      | F      | G |
|---|---|--------|--------|--------|--------|--------|---|
| 1 |   |        |        |        |        |        |   |
| 2 |   |        |        |        |        |        |   |
| 3 |   | Wert 1 | Wert 2 | Wert 3 | Wert 4 | Wert 5 |   |
| 4 |   | 12     | 343    | 121    | 77     | 37     |   |
| 5 |   | 96     | 28     | 359    | 155    | 745    |   |
| 6 |   | 258    | 23     | 114    | 934    | 126    |   |
| 7 |   | 425    | 749    | 248    | 90     | 83     |   |
| 8 |   | 53     | 367    | 238    | 482    | 142    |   |
| 9 |   |        |        |        |        |        |   |

## Beispiel 1: Bei Doppelklick auf eine Zelle, den Zellentext rot färben

```
Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel As Boolean)

    'Textfarbe bei Doppelklick ändern
    If Target.Font.Color = vbRed Then
        'Von Rot auf Schwarz ändern
        Target.Font.Color = vbBlack
    Else
        'Von Schwarz auf Rot ändern
        Target.Font.Color = vbRed
    End if

    'Bearbeitungsmodus deaktivieren
    Cancel = True

End Sub
```

## Beispiel 2: Bei Rechtsklick auf eine Zelle, den Zelleninhalt kopieren

```
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)

    'Zelleninhalt in die Zwischenablage kopieren
    MsgBox "Zelleninhalt in die Zwischenablage kopiert"
    Target.Copy
    'Kontextmenu deaktivieren
    Cancel = True

End Sub
```



## 30 Tabellenblattereignisse (Change, BeforeDoubleClick, ...)

**Beispiel 3:** Nur bestimmte Eingaben in die Zellen erlauben

```
Private Sub Worksheet_Change(ByVal Target As Range)

    'In Spalte B dürfen nur Zahlen eingegeben werden
    If Target.Column = 2 Then

        If Not IsNumeric(Target) Then
            'Keine Zahl wurde eingetragen
            MsgBox "Bitte geben Sie eine Zahl an"
            'Eintrag nicht übernehmen
            Target.ClearContents
        End if

    End if

End Sub
```